A MOS Controlled Thyristor Model for PSpice Simulation

Goce L. Arsov, Ph.D. and Ljupco P. Panovski, Ph.D.

Faculty of Electrical Engineering, St. Cyril and Methodius University, 91000 Skopje, P.O.Box 574, Republic of Macedonia

Abstract - A PSpice model for the MOS controlled thyristor (MCT) consisting of a thyristor section and two MOSFET sections is proposed. The parameters of the equivalent circuit can be calculated from the data provided by the manufacturer and some simple measurements. The simulation results of a phase controlled rectifier show quite accurate behaviour of the proposed model.

Keywords - PSpice, simulation, modeling, MOS controlled thyristor, power semiconductor

1 INTRODUCTION

The MOS Controlled Thyristor (MCT) is a new power semiconductor device having several advantages compared with other devices. The MCT has a main thyristor section, and a gate control section for turning it on and off. Since the integration of the MCT is complex, it is very difficult to obtain an exact circuit model for the device. A complex model of the MCT has been developed by Harris Semiconductor [1]. A simpler model for the MCT would help to keep the simulation time and effort minimal. Such a model has been proposed in [2], but it seems that it does not simulate all the MCT characteristics properly (such as: SCR breakdown and breakover, turn-on spike voltages...).

This paper presents a simple circuit model for the P-type MCT suitable for analysis of MCT power converters. The model represents the characteristics of the MCT during turn-on and turn-off, as well as the breakdown and breakover characteristics of the SCR section, including their temperature dependence. The model parameters can be obtained from the specifications contained in the manufacturer's data sheet as described in [2], [3] and [4]. Extensive simulations have been performed in order to test the proposed model characteristics. The model has passed all performed tests and can be used to analyze various types of MCT power converters. The simulation results of an MCT phase-controlled rectifier are shown below.

2 CIRCUIT MODEL FOR THE MCT

The PSpice model for the MCT is derived from the transistor level equivalent of MCT shown in Fig. 1. Its SCR section is developed by expanding the thyristor model presented in [3] and [4]. The gate control section consisting of a PMOS and an NMOS is modeled using simple RC circuits for the controlling of the turn-on and turn-off times and with simple diodes (D_{PMOS} and D_{NMOS}) to isolate the operation of the MOSFET's in such a manner that the triggering pulse of one polarity activates only one FET.



Fig. 1 Transistor level equivalent circuit of the MCT

The complete PSpice model is shown in Fig. 2 and the complete list of the PSpice statements for the subcircuit description of the proposed MCT model is presented in the appendix.

When a negative pulse is applied to the gate of the PMOS, the capacitor C_{PMOS} charges up. The voltage across C_{PMOS} increases the value of G_P which charges the capacitor C_P. The current through V_{GD} causes the positive current from F_{CTROL} and a negative voltage V(3,20) appears between nodes 3 and 20. The voltage across E_3 will be zero and, as a consequence, the voltage across E_1 will be also zero. The MCT is in conducting state. Once the MCT is turned on, the current through V_5 will keep the device in the on state.

Once a positive pulse is applied to the gate, the capacitor C_{NMOS} charges, the G_P reverses its direction and so does the current from F_{CTROL}. The voltage across E_3 rises to 1 and the voltage across E_1 takes the value of the anode to cathode voltage. The MCT is in the off state. The behaviour of the SCR section of the model is explained in [3] and [4].



Fig. 2 PSpice model of the MCT

3 SIMULATION RESULTS

Extensive simulations have been performed in order to test the proposed model characteristics. The model has passed all performed tests and can be used to analyze various types of MCT power converters. The simulation results for the complete circuit of an MCT phase-controlled rectifier, shown in Fig. 3, are presented below. The converter uses a diode bridge to rectify the input ac voltage. The MCT is used as a phase-controlled switch to control the flow of power from the ac source. The diode D_m provides the freewheeling action. The waveforms of the input voltage, load voltage and gate pulses are shown in Fig. 4. Comparing them with the experimental waveforms shown in [2] it can be seen that the simulated waveforms agree very well with the experimental ones.



Fig. 3 Power circuit of the MCT phase-controlled rectifier



Fig. 4 Simulated MCT converter waveforms

4 CONCLUSION

The paper presents a PSpice model of the MCT. The model has a thyristor section and two MOSFET sections for turning the MCT on and off. The model is simple and it represents the characteristics of the MCT during turn-on and turn-off, as well as the breakdown and breakover characteristics of the SCR section, including their temperature dependence. The extensive simulations performed show very good behaviour of the proposed model .

REFERENCES:

1. MCT User`s Guide, Harris Semiconductor Corporation, 1995

2. S. Yuvarnijan and D. Quek, "A PSPICE Model for the MOS Controlled Thyristor" *IEEE Trans. Ind. Electron*, Vol. 42, No 5, pp 554-558, 1995.

3. F. J. Gracia, F. Arizti, and F.J. Aranceta "A Nonideal Macromodel of Thyristor for TransientAnalysis in Power Electronic Systems", *IEEE Trans. Ind. Electron*, Vol. 37, No 6, pp 514-520, 1990.

4. G. L. Arsov "Comments on 'A Nonideal Macromodel of Thyristor for TransientAnalysis in Power Electronic Systems'", *IEEE Trans. Ind. Electron*, Vol. 39, No 2, pp 175-176, 1992.

APPENDIX

Subcircuit description of the MCT

```
*PSPICE MODEL FOR MCT
.SUBCKT MCT 10 20 30
*----------------A K G

*PNPN STRUCTURE
.MODEL DMOD D
.MODEL DON1 D (RS=.019 IS=1E-12)
.MODEL DGAT D (IS=1E-12)
.MODEL DBREAK D (BV=900 IS=1E-7 XTI=3)
.MODEL DBREAK1 D (BV=20 IS=1E-7 XTI=3)
V1 1 20 DC .999
V5 10 14 DC 0
VREV 65 14 DC 0
D1 3 1 DMOD
D2 20 3 DMOD
D3 67 56 DMOD
D4 56 70 DMOD
D5 20 56 DMOD
DON 14 22 DON1
DBREAK 14 27 DBREAK1
D6 20 27 DBREAK
E1 22 20 POLY(2) 10 20 5 20 0 0 0 0 1
E2 66 20 10 20 -1
C1 3 20 .5UF IC=1.5
CRISE 14 20 1NF
COFF 56 20 10NF
RLEAK 10 20 40K
ROFF 66 67 2.5
R10 66 70 1
GOFF 20 65 66 67 1
FCTROL 3 20 POLY (3) VGD V5 VREV -.1 1 .5 1
E3 5 20 TABLE {V(3,20)}=(.01V,0V) (1.25V,1V)
R3 5 20 10MEG

*MOS STRUCTURE
RGP   83   84   5
RGN   85   95   5
CPMOS   10   83   18.92N
DPMOS   84   30   DPN
CNMOS   85   10   31.54N
DNMOS   30   95   DPN
RDP   10   84   100
RDN   10   95   100
.MODEL   DPN   D
GP 93 0 POLY(2) 10 83 10 85 0 6.423 32.15
RP   0   92   1
VGD   92   93   0
CP   0   93   .07U
.ENDS
```